



Coordonnées polaires

(Partie 1 de 2)

© Pierre Lantagne

Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue en février 2001. Ce document est le premier d'une série de deux portant sur les coordonnées polaires. L'objectif de la première partie est de détailler le système de coordonnées polaires. Après avoir défini ce qu'est le système de coordonnées polaires, nous présenterons les équations de transformation du système polaire au système cartésien. Nous illustrerons ensuite la représentation graphique de quelques lieux géométriques remarquables définies en coordonnées polaires avec un quadrillage polaire et avec un quadrillage cartésien. Pour terminer ce premier document, nous verrons les précautions à prendre dans la résolution de systèmes d'équations qui sont à poser lorsqu'il faut déterminer les points d'intersection de courbes définies en coordonnées polaires.

Quant à l'objectif de la seconde partie, c'est celui de montrer la manière dont s'effectue le tracé d'une courbe définie en coordonnées polaires. On y trouvera une procédure permettant d'animer, étape par étape, la création d'une courbe définie en coordonnées polaires permettant de bien voir **le début du tracé** de la courbe ainsi que **le sens dont s'effectue le tracé**.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.2

Initialisation


```
> restart;
> with(plots, coordplot, display, polarplot, setoptions, textplot):
  setoptions(font=[TIMES, ROMAN, 8], axesfont=[TIMES, ROMAN, 8], labelfont=
    [TIMES, ROMAN, 8], titlefont = ["ROMAN", 12]);
  with(plottools, disk, rotate, translate):
  Fond:=ColorTools:-Color([1,0.99215686,0.96862745]):
```

Plan polaire

Le plan polaire est le plan géométrique muni d'un repère polaire. Un repère polaire est composé d'une origine appelée **pôle** d'où est issue une demi-droite horizontale (orientée à droite) graduée appelée **axe polaire**.

```
> Axe_polaire:=plot([[0,0],[4,0]],thickness=2,color=orange):
  Pôle:=disk([0,0], .05, color=orange):
  Quadrillage_polaire:=coordplot(polar,[0..4,0..2*Pi],
    labelling=front,
    grid=[5,9],
    color=[grey, grey],
    linestyle = [1, 1],
    scaling=constrained):
  display(Quadrillage_polaire, Axe_polaire, Pôle, view=[-4..4, -4..4],
```

```
title=typeset("Quadrillage polaire");
```




Un point $P(r, \theta)$ dont les coordonnées polaires sont $r \in \mathbb{R}$ et $\theta \in \mathbb{R}$ est repéré de la manière suivante

- r est la *distance orientée* à parcourir sur l'axe polaire selon le sens de l'axe polaire et
- θ est l'*angle orienté* à parcourir selon le sens trigonométrique.

Illustrons le point de coordonnées polaires $P\left(2, \frac{\pi}{3}\right)$ dans le plan polaire en exécutant les requêtes du bloc ci-dessous.


```
> Axe_polaire:=plot([[0,0],[4,0]],thickness=2,color=orange):
Pôle:=disk([0,0], .03, color=orange):
Arc:=plot([2*cos(theta),2*sin(theta),theta=0..Pi/3],thickness=1,
linestyle=2,color=navy):
Rayon:=plot([[0,0],[2*cos(Pi/3),2*sin(Pi/3)]],color=orange,thickness=
1,linestyle=2):
PointP:=disk([2*cos(Pi/3),2*sin(Pi/3)], .02, color=black):
Texte_0:=textplot([0,-0.05,typeset(0)],font=[TIMES,ROMAN,12],color=
orange,align={BELOW}):
Texte_1:=textplot([0.6,1,typeset(abs(r)=2)],font=[TIMES,ROMAN,12],
color=orange,align={LEFT}):
Texte_2:=textplot([0.2,0.1,typeset(theta=Pi/3)],font=[roman,12],
color=navy,align={ABOVE,RIGHT}):
Texte_3:=textplot([2*cos(Pi/3)+.1,2*sin(Pi/3),typeset(P(2,Pi/3))],
font=[roman,12],color=navy,align={RIGHT}):
#Texte_4:=textplot([2*cos(Pi/3)+0.04,2*sin(Pi/3),`P`],font=[TIMES,
ROMAN,12],color=navy,align={ABOVE,RIGHT}):
Éléments:=[Pôle,Axe_polaire,Arc,Rayon,PointP,Texte_| |(0..3)],scaling=
constrained:
```

```
display(Éléments,axes=None);
```




Illustrons le point P du plan polaire avec un quadrillage polaire.

```
> PointP:=disk([2*cos(Pi/3),2*sin(Pi/3)], .08, color=black):
  Texte_3:=textplot([2*cos(Pi/3)+.1,2*sin(Pi/3)],typeset("P"),font=
    [roman,12],color=navy,align={ABOVE,RIGHT}):
> display([Quadrillage_polaire,Texte_3,Pôle,Axe_polaire,PointP]);
```



Nous pouvons tout aussi bien représenter ce point dans un repère cartésien bien que les coordonnées du point P soient polaires.


```
> display(Éléments,axes=normal,view=[-2..2,-2..2]);
```



Avec un repère polaire, représentons les points $Q\left(-2, \frac{\pi}{3}\right)$, $R\left(2, -\frac{\pi}{4}\right)$ et $S\left(-1, -\frac{3\pi}{4}\right)$.

Il s'agit de représenter des points dont les coordonnées polaires r et θ sont aussi bien des valeurs positives que négatives. Il faut d'abord positionner l'axe polaire dans la direction de la coordonnée angulaire θ tout en respectant le sens trigonométrique positif ou négatif. Ensuite, si la coordonnée r est positive, de dépasser d'une distance r unités dans le sens de l'axe polaire mais si la coordonnée r est **négative**, se déplacer d'une distance $|r|$ unités **dans le sens opposé** de l'axe polaire.

```
> PointQ:=disk([-2*cos(Pi/3),-2*sin(Pi/3)], .07, color="Niagara 1"):
Texte_Q:=textplot([-2*cos(Pi/3)-.20,-2*sin(Pi/3),typeset(Q)],font=
[TIMES,BOLD,18],color=navy,align={above,left}):
PointR:=disk([2*cos(-Pi/4),2*sin(-Pi/4)], .07, color="Niagara 1"):
Texte_R:=textplot([2*cos(-Pi/4)+.2,2*sin(-Pi/4),typeset(R)],font=
[TIMES,BOLD,18],color=navy,align={right}):
PointS:=disk([-cos(-3*Pi/4),-sin(-3*Pi/4)], .07, color="Niagara 1"):
Texte_S:=textplot([-cos(-3*Pi/4)+.1,-sin(-3*Pi/4),typeset(S)],font=
[TIMES,BOLD,18],color=navy,align={right}):
display([Quadrillage_polaire,Pôle,Axe_polaire,PointQ,PointR,PointS,
Texte_Q,Texte_R,Texte_S]);
```



Notons qu'un même point dans un repère polaire possède une infinité d'écritures en coordonnées polaires puisqu'il est possible de spécifier une même direction avec une infinité de coordonnée angulaire θ différentes et d'ajuster en conséquence le signe de la première coordonnée r .

Relation entre coordonnées cartésiennes et coordonnées polaires

Si nous faisons coïncider le pôle du repère polaire avec l'origine du repère cartésien, les points de coordonnées polaires (r, θ) ont des coordonnées cartésiennes (x, y) établies par les relations suivantes:

$$x = r \cos(\theta)$$

$$y = r \sin(\theta)$$

Et réciproquement, le plan cartésien est le plan géométrique dont les points de coordonnées cartésiennes (x, y) ont des coordonnées polaires (r, θ) établies par les relations suivantes:

$$r^2 = x^2 + y^2$$

$$\tan(\theta) = \frac{y}{x}$$

La coordonnée r représente la *distance orientée* à parcourir sur l'axe polaire. Lorsque la coordonnée r du point $P(r, \theta)$ est négative, la distance séparant le pôle et le point P est $|r|$ et cette distance est parcourue dans le sens opposé de l'axe polaire. Ainsi, le côté terminal OP est celui d'un angle de $\theta + \pi$. Ainsi, les équations

$$\cos(\theta + \pi) = \frac{x}{r} \text{ et } \sin(\theta + \pi) = \frac{y}{r}$$

se ramènent également aux équations

$$x = r \cos(\theta) \text{ et } y = r \sin(\theta)$$

Dans tous les cas, les équations $x = r \cos(\theta)$ et $y = r \sin(\theta)$ serviront à déduire la forme cartésienne correspondant la forme polaire d'une équation et vice-et-versa.

Trouvons l'équation ou les équations polaires correspondant à l'équation cartésienne $x^2 - y^2 = a^2$. On

reconnait cette équation comme étant celle d'une famille d'hyperbole. Nous allons donc obtenir la forme polaire de cette famille.

$$\begin{aligned} > \text{Éq1} := x^2 - y^2 = a^2; \\ & \qquad \qquad \qquad \text{Éq1} := x^2 - y^2 = a^2 \end{aligned} \tag{3.1}$$

$$\begin{aligned} > \text{Éq2} := \text{subs}(x=r*\cos(\theta), y=r*\sin(\theta), \text{Éq1}); \\ & \qquad \qquad \qquad \text{Éq2} := r^2 \cos^2(\theta) - r^2 \sin^2(\theta) = a^2 \end{aligned} \tag{3.2}$$

$$\begin{aligned} > \text{Sol} := \text{solve}(\text{Éq2}, \{r\}); \\ & \qquad \qquad \qquad \text{Sol} := \left\{ r = -\frac{a}{\sqrt{1-2\sin^2(\theta)}} \right\}, \left\{ r = \frac{a}{\sqrt{1-2\sin^2(\theta)}} \right\} \end{aligned} \tag{3.3}$$

$$\begin{aligned} > \text{simplify}(\text{Sol}[1], [(\sin^2)(\theta) = (1 - \cos(2*\theta))/2]); \\ & \qquad \qquad \qquad \text{simplify}(\text{Sol}[2], [(\sin^2)(\theta) = (1 - \cos(2*\theta))/2]); \\ & \qquad \qquad \qquad \left\{ r = -\frac{a}{\sqrt{\cos(2\theta)}} \right\} \\ & \qquad \qquad \qquad \left\{ r = \frac{a}{\sqrt{\cos(2\theta)}} \right\} \end{aligned} \tag{3.4}$$

Voilà !

Comme second exemple, trouvons l'équation cartésienne correspondant à l'équation polaire $r = a \operatorname{cosec}(\theta)$.

$$\begin{aligned} > \text{Éq1} := r = a * \operatorname{csc}(\theta); \\ & \qquad \qquad \qquad \text{Éq1} := r = a \operatorname{csc}(\theta) \end{aligned} \tag{3.5}$$

$$\begin{aligned} > \text{Éq2} := \text{subs}(\operatorname{csc}(\theta) = r/y, r = x^2 + y^2, \text{Éq1}); \\ & \qquad \qquad \qquad \text{Éq2} := x^2 + y^2 = \frac{a(x^2 + y^2)}{y} \end{aligned} \tag{3.6}$$

$$\begin{aligned} > \text{Éq3} := \text{Éq2} / (x^2 + y^2); \\ & \qquad \qquad \qquad \text{Éq3} := 1 = \frac{a}{y} \end{aligned} \tag{3.7}$$

$$\begin{aligned} > \text{Sol} := \text{solve}(\text{Éq3}, \{y\}); \\ & \qquad \qquad \qquad \text{Sol} := \{y = a\} \end{aligned} \tag{3.8}$$

Nous constatons finalement que l'équation polaire $r = a \operatorname{cosec}(\theta)$ représente dans le plan cartésien la famille de droites horizontales d'équation $y = a$.

Voyons maintenant la manière de s'y prendre pour obtenir des tracés de lieux géométriques avec des équations polaires.

Graphiques d'équations polaires

Dans le plan cartésien, une équation de la forme $y = f(x)$ décrit un certain lieu géométrique. Une équation polaire de la forme $r = f(\theta)$ décrira donc, dans le plan polaire, un certain lieu géométrique également.

Le tracé de courbes d'équations polaires de la forme $r = f(\theta)$ peut être obtenu avec la macro-

commande `plot` de la bibliothèque principale en spécifiant, en option, le système de coordonnées `coords=polar`. Dans ce cas, l'afficheur superposera tous les tracés avec un repère cartésien plutôt qu'avec un repère polaire.

La **syntaxe paramétrique** de la macro-commande `plot` est interprétée par le simplificateur comme suit:

- le premier argument de la liste est une fonction de t donnant la valeur du rayon r
- le second argument de la liste est une fonction de t donnant la valeur de l'angle θ

```
plot([r(t),theta(t),t=a..b],h,v,coords=polar,options)
```


Alors, pour le tracé de courbes d'équations polaires de la forme $r = f(\theta)$, il suffit de poser

- la fonction r comme une fonction du paramètre t : $r(t) = \text{rayon}$
- la fonction θ comme la fonction identité $\theta(t) = t$.

```
plot([r(t),t,t=a..b],h,v,coords=polar,options)
```

Premier exemple: dans le plan polaire, traçons le lieu d'équation $r = 1$ pour $\theta \in [0, 2\pi]$. On devrait obtenir évidemment un cercle centré à l'origine de rayon unité.


```
> plot([1,theta,theta=0..2*Pi],-1..1,-1..1,  
      coords=polar,  
      color=navy,  
      thickness=2,  
      scaling=constrained);
```



Second exemple: dans le plan polaire, traçons le lieu d'équation $r = 2 \sin(\theta)$ pour $\theta \in [0, \pi]$. Là, le lieu géométrique qui sera tracé est moins évident à deviner.

```
> plot([2*sin(theta),theta,theta=0..Pi],-1.5..1.5,-0.5..2.5,  
      coords=polar,  
      color=navy,
```

```
thickness=2,  
scaling=constrained);
```




Remarquons que le contrôle de l'affichage des deux graphiques précédents **n'a pas été fait** avec l'option `view`. Cette façon est propre à la syntaxe paramétrique ([plot,parametric](#)). L'option passe-partout `view` aurait pu tout aussi bien faire l'affaire bien sûr.

La macro-commande [polarplot](#) de la bibliothèque `plots` fait de manière plus conviviale le tracé d'équations polaires de la forme $rayon = r(\theta)$.

```
polarplot(r(theta), theta=alpha..beta, options)
```


Notons ici que l'évaluateur considère automatiquement le premier argument $r(\theta)$ comme étant la fonction donnant la valeur du rayon.

```
> PolarOptions:=axiscoordinates=polar,  
axis[radial]=[tickmarks=10],  
axis[angular]=[tickmarks=24],  
thickness=3,  
color="Niagara 1":  
> polarplot(1,theta=0..2*Pi,PolarOptions,coordinateview=[-1.25..1.25,0.  
.2*Pi]);
```




REMARQUE: Si l'intervalle de valeurs d'angles θ n'est pas spécifié dans la macro-commande `polarplot`, alors **implicitement**, l'intervalle sera par défaut l'intervalle $0 \leq \theta \leq 2\pi$.

```
> polarplot(2*sin(theta),PolarOptions);
```




```
> polarplot(2*sin(theta),theta=0..-Pi/2,PolarOptions);
```




Avec la macro-commande `polarplot`, on ne peut de contrôler l'affichage avec l'option `view`.

```
> polarplot(2*sin(theta),
            PolarOptions,
            view=[-3..3,0..2*Pi]);
```



Mais l'on peut avec l'option `Coordinateview`. lorsque `axiscoordinates` a été initialisée à la valeur `polar` (Voir `OptionsPolar`)


```
> polarplot(2*sin(theta),
            PolarOptions,
            coordinateview=[-2.5..2.5,0..2*Pi]);
```



La macro-commande `polarplot` est accompagnée de son propre mode d'affichage sur lequel l'utilisateur dispose d'un certain contrôle (faites un clic-droit sur le graphique précédent et allez à Axes → Propriétés... pour une gestion du tracé.

Pour disposer de l'option `View`, on doit utiliser `plot` avec l'option `coords=polar`.

```
> plot([2*sin(theta),theta,theta=0..-Pi/2],
        coords=polar,
        OptionsPolar,
        view=[-1.5..1.5,-0.5..2.5]);
```



Dans un cas comme dans l'autre, `plot` avec l'option `coords=polar` et `polarplot` ne permettent pas le tracé

de la réciproque $\theta = f^{-1}(r)$ ni le tracé d'équations de la forme $r^2 = f(\theta)$.

Il y a une autre forme de la macro-commande `polarplot` qui permet le tracé d'équations polaires autres que celles de la forme $r = f(\theta)$.


```
polarplot([f(t), g(t), t =  $\alpha$ .. $\beta$ ], a..b, c..d, options)
```

Cette forme est la **forme paramétrique polaire** de la macro-commande `polarplot`. L'évaluateur traite

- le premier terme de la liste est la formule qui précise le rayon r en fonction du paramètre t : $r = f(t)$
- le deuxième terme de la liste est la formule qui précise l'angle θ en fonction du paramètre t : $\theta = g(t)$
- le troisième terme de la liste est l'intervalle de nombres réels (radians) dans lequel variera le paramètre t .


L'exemple suivant illustrera la différence entre la syntaxe paramétrique cartésienne et la syntaxe paramétrique polaire.

```
> A:=plot([cos(t),sin(t),t=0..2*Pi],  
          title=typeset("Paramétrisation cartésienne \n",x=cos  
(theta),"\n" ,y=sin(theta)),  
          thickness=2,axes=None,  
          color="Niagara 8");  
B:=polarplot([cos(t),sin(t),t=0..2*Pi],  
             title=typeset("Paramétrisation polaire\n", r=cos(t),"\n",  
theta=sin(t)),  
             thickness=2,axiscoordinates=cartesian,axes=None,  
             color="Niagara 8");  
display(Matrix(1,2,[A,B]),scaling=constrained);
```



L'exemple suivant vous sera peut-être utile à l'occasion de la Saint-Valentin.


```
> Coeur:=polarplot([-abs(sqrt(t)/2),((t-Pi)/2),t=-2*Pi..2*Pi],
    color="Niagara 1",
    transparency=0,
    filled=true,
    axes=None);
display(translate(Coeur,0,0.5),size=[300,300]);
```



Les équations polaires explicites de la forme $r = f(\theta)$ peuvent être aussi tracées avec la syntaxe paramétrique polaire en prenant pour fonction g la fonction identité $g(t) = t$.


```
polarplot([f(t),t,t=α..β],a..b,c..d,options)
```

```
> polarplot([1,theta,theta=0..2*Pi],
    color="Niagara 5",
    thickness=2,
    coordinateview=[-1.5..1.5,0..2*Pi]);
```



```
> polarplot([2/sqrt(cos(2*theta)),theta,theta=-2*Pi..2*Pi],
    color="Niagara 13",
```

```
thickness=2,
coordinateview=[-20..20,0..2*Pi]);
```



La droite

Avec une équation linéaire de la forme $ax + by + c = 0$, les équations de transformation $x = r \cos(\theta)$ et $y = r \sin(\theta)$ nous conduisent à l'équation

$$r = -\frac{c}{a \cos(\theta) + b \sin(\theta)}$$

En effet,

```
> Éq:=a*x+b*y+c=0;
```

$$\dot{Éq} := ax + by + c = 0 \quad (5.1)$$

```
> isolate(subs(x = r*cos(theta), y = r*sin(theta), Éq), r);
```

$$r = -\frac{c}{a \cos(\theta) + b \sin(\theta)} \quad (5.2)$$

Obtenons l'équation polaire de la droite d'équation cartésienne $2x + 3y + 2 = 0$. Traçons ensuite la courbe décrite par cette équation polaire.

```
> Éq1:=2*x+3*y+2 = 0;
```

$$\dot{Éq1} := 2x + 3y + 2 = 0 \quad (5.3)$$

```
> Éq2:=subs(x=r*cos(theta), y=r*sin(theta), Éq1);
```


$$\dot{Éq2} := 2r \cos(\theta) + 3r \sin(\theta) + 2 = 0 \quad (5.4)$$

```
> Éq_polaire:=isolate(Éq2, r);
```

$$\dot{Éq_polaire} := r = -\frac{2}{2 \cos(\theta) + 3 \sin(\theta)} \quad (5.5)$$

```
> polarplot([rhs(Éq_polaire), theta, theta=0..Pi],
color="Niagara 13",
```

```
thickness=2,  
coordinateview=[-5..5,0..2*Pi]);
```




Lorsque $c = 0$, la droite passe alors par l'origine. Alors, de telles droites **ne peuvent** s'exprimer par la forme

$$r = -\frac{c}{a \cos(\theta) + b \sin(\theta)}.$$

Dans le plan polaire, les droites passant par le pôle sont évidemment de la forme $\theta = \text{constante}$. Il suffit donc de connaître leur inclinaison θ .

Traçons la droite d'équation polaire $\theta = -\frac{\pi}{6}$.

```
> polarplot([t,-Pi/6, t=-10..10]);
```



Traçons la droite d'équation $x - 6y = 0$.

```
> Éq:=x-6*y=0;
```


$$\acute{E}q := x - 6y = 0 \quad (5.6)$$

```
> isolate(Éq,y);
```

$$y = \frac{x}{6} \quad (5.7)$$

Puisque $m = \frac{1}{6}$, alors $\theta = \arctan\left(\frac{1}{6}\right)$.

```
> polarplot([t,arctan(1/6), t=-10..10]);
```



Dans les cas de droites verticales d'équations $x = a$ et des droites horizontales d'équations $y = b$, la transformation est directe:

$$r \cos(\theta) = a \text{ et } r \sin(\theta) = b$$


REMARQUE: il faudra tenir compte des valeurs de discontinuité.

– droites verticales: $r = \frac{a}{\cos(\theta)} = a \sec(\theta)$

– droites horizontales: $r = \frac{a}{\sin(\theta)} = a \csc(\theta)$

Traçons les droites d'équation $x = 3$ et d'équation $y = \frac{3}{2}$.

```
> D1:=polarplot(3*sec(theta),theta=-Pi/3..Pi/3,
  title=typeset("Droite d'équation cartésienne ",x=3,"\n et
  donc d'équation polaire ",r=3*sec(theta) ),
  color="Niagara 2",
  thickness=3,
  coordinateview=[-5..5,0..2*Pi]):
D2:=polarplot(3/2*csc(theta),theta=-Pi..Pi,
  title=typeset("Droite d'équation cartésienne ",y=3/2, "\n et
  donc d'équation polaire ",r=3/2*csc(theta)),
  color="Niagara 2",
  thickness=3,
  coordinateview=[- 3..3,0..2*Pi]):
display(Matrix(1,2,[D1,D2]),titlefont = ["ROMAN", 15]);
```



▼ Le cercle

L'équation cartésienne $x^2 + y^2 = a^2$ se transforme en équation polaire $r^2 = a^2$. En effet,

$$\begin{aligned} > \text{Éq1} := x^2 + y^2 = a^2; \\ & \qquad \qquad \qquad \text{Éq1} := x^2 + y^2 = a^2 \end{aligned} \tag{6.1}$$


$$\begin{aligned} > \text{Éq2} := \text{subs}(x=r*\cos(\theta), y=r*\sin(\theta), \text{Éq1}); \\ & \qquad \qquad \qquad \text{Éq2} := r^2 \cos^2(\theta) + r^2 \sin^2(\theta) = a^2 \end{aligned} \tag{6.2}$$

$$\begin{aligned} > \text{Éq3} := \text{simplify}(\text{Éq2}); \\ & \qquad \qquad \qquad \text{Éq3} := r^2 = a^2 \end{aligned} \tag{6.3}$$

$$\begin{aligned} > \text{solve}(\text{Éq3}, \{r\}); \\ & \qquad \qquad \qquad \{r = a\}, \{r = -a\} \end{aligned} \tag{6.4}$$

Nous avons donc les équations polaires $r = a$ ou $r = -a$. Avec des intervalles d'angles θ appropriés, ces deux équations représentent, dans le plan cartésien, le **même cercle** centré à l'origine.

```
> A:=polarplot(2,theta=0..2*Pi,
    color="Niagara 13",
    thickness=2,
    title=typeset(r=2)):
B:=polarplot(-2,theta=0..2*Pi,
    color="Niagara 13",
    thickness=2,
    title=typeset(r=-2)):
display(Matrix(1,2,[A,B]),coordinateview=[-3..3,0..2*Pi]);
```



Ces deux cercles sont identiques mais la manière dont leur tracé a été effectué est différente. Prendre connaissance de la seconde partie « Coordonnées polaires II » pour analyser dans le détail cette affirmation.

Obtenons l'équation polaire d'un cercle d'équation cartésienne $(x - a)^2 + y^2 = a^2$. On reconnaît que cette équation représente la famille de cercles de rayon $|a|$ centrés au point $C(a,0)$.

```
> Éq1:=(x-a)^2+y^2=a^2;
Éq2:=subs(x=r*cos(theta),y=r*sin(theta),Éq1);
Éq_polaire:=solve(Éq2,{r});
```

$$\begin{aligned} \text{Éq1} &:= (x - a)^2 + y^2 = a^2 \\ \text{Éq2} &:= (r \cos(\theta) - a)^2 + r^2 \sin(\theta)^2 = a^2 \\ \text{Éq_polaire} &:= \left\{ r = 0 \right\}, \left\{ r = \frac{2a \cos(\theta)}{\cos(\theta)^2 + \sin(\theta)^2} \right\} \end{aligned} \quad (6.5)$$


La solution Éq_polaire_1 , soit $r = 0$ est à rejeter.

```
> simplify(Éq_polaire[2]);
```

$$\{r = 2a \cos(\theta)\} \quad (6.6)$$

Comme exemple, traçons le lieu d'équation polaire $r = 2a \cos(\theta)$ en prenant $a = 2$.

```
> polarplot(4*cos(theta),theta=-Pi..Pi,thickness=2);
```



Il s'agit effectivement d'un cercle de rayon 2 centré au point $C(2,0)$.

Dans le plan polaire, une équation polaire de la forme

$r = a$	est celle d'un cercle de centre: $(0,0)$ et de rayon a
$r = 2a \cos(\theta)$	est celle d'un cercle de centre: $(a,0)$ et de rayon a
$r = 2a \sin(\theta)$	est celle d'un cercle de centre: $(0,a)$ et de rayon a

Les rotations en coordonnées polaires


Le tracé de l'équation polaire $r = f(\theta - \alpha)$ est celui de l'équation polaire $r = f(\theta)$ ayant subi une rotation d'un angle α :

- si l'angle α est positif, le tracé initial subit une rotation dans le sens anti-horaire, et

- si l'angle α est négatif, la rotation est effectuée dans le sens horaire.

Dans un même graphique, superposons le tracé du cercle précédent et celui de sa rotation d'un angle $\alpha = 90$ degrés.

```
> Cercle:=polarplot(4*cos(theta),theta=-Pi..Pi,thickness=2):
Cercle_Rot:=polarplot(4*cos(theta-Pi/2),theta=-Pi..Pi,color="Niagara
16",thickness=2):
> display({Cercle,Cercle_Rot},title=typeset("Rotation anti-
horaire"),coordinateview=[-5..5,0..2*Pi]);
```



Résumons. Une équation polaire de la forme $r = 2a \cos(\theta - \alpha)$ décrit, dans le plan polaire, un cercle de rayon a centré au point $C(a, \alpha)$

Options de traçage

À l'aide de la macro-commande `setoptions` de la bibliothèque `plots`, rendons globales les options suivantes:

```
color= "Niagara 12"
thickness=2
```

```
> setoptions(color="Niagara 6",thickness=2):
```


Cardioides et limaçons

Formes générales: $r = a + b \cos(\theta)$ et $r = a + b \sin(\theta)$


En général:

```
|a| >= |b| cardioïde
|a| < |b| limaçon
```


```
> polarplot(2+2*cos(theta),theta=0..2*Pi);
```




```
> polarplot(-2+2*cos(theta),theta=0..2*Pi);
```



```
> polarplot(1+2*cos(theta),theta=0..2*Pi);
```




```
> polarplot(3+2*cos(theta),theta=0..2*Pi);
```




Rosaces

Forme générale: $r = a \cos(n\theta)$ et $r = a \sin(\theta)$


```
> polarplot(4*cos(2*theta),theta=0..2*Pi);
```



```
> polarplot(5*sin(4*theta),theta=0..2*Pi);
```




```
> polarplot(4*cos(2*(theta+Pi/4)),theta=0..2*Pi);
```



Lemniscate

Forme générale: $r^2 = a^2 \cos(2\theta)$. Avec la macro-commande `polarplot`, nous devons tracer ce lieu géométrique en deux parties.

```
> A:=polarplot(12*cos(2*theta),theta=-Pi/4..Pi/4):
  B:=polarplot(-12*cos(2*theta),theta=-Pi/4..Pi/4):
> display(A,B);
```




Intersection de courbes en coordonnées polaires

On conviendra d'exprimer la coordonnée polaire $\theta \in [0, 2\pi[$.

Exemple 1

Soit les cercles d'équations polaires $r = 4 \sin(\theta)$ et $r = 4 \cos(\theta)$. Trouvons les coordonnées des points d'intersection de ces deux cercles.

```
> C1:=polarplot(4*sin(theta),theta=0..Pi):
  C2:=polarplot(4*cos(theta),theta=0..Pi,color="Niagara 15"):
> display([C1,C2]);
```



Il y a seulement deux points d'intersection.

```
> Éq_1:= r=4*sin(theta);
  Éq_2:= r=4*cos(theta);
```

$$\dot{Éq}_1 := r = 4 \sin(\theta)$$

$$\dot{Éq}_2 := r = 4 \cos(\theta) \tag{12.1.1}$$

```
> solve({Éq_1,Éq_2},{r,theta});
```

$$\left\{ r = 2\sqrt{2}, \theta = \frac{\pi}{4} \right\} \tag{12.1.2}$$

Le graphique précédent montre pourtant que l'origine est aussi un point d'intersection.

Tentons, en initialisant la variable d'environnement `_EnvAllSolutions`, d'obtenir toutes les solutions.

```
> _EnvAllSolutions:=true;
```

$$_EnvAllSolutions := true \tag{12.1.3}$$

```
> solve({Éq_1,Éq_2},{r,theta});
```

$$\left\{ r = 4 \cos\left(\frac{1}{4} \pi + \pi_{ZI}\right), \theta = \frac{1}{4} \pi + \pi_{ZI} \right\} \tag{12.1.4}$$

Le domaine de la variable système `_ZI` est l'ensemble des entiers \mathbb{Z} . L'origine (le pôle) n'a pas été obtenu par la résolution simultanée des deux équations polaires. La valeur $r = 0$ ne peut être obtenue dans les deux équations simultanément par un même angle θ . On doit donc s'inspirer des tracés des deux graphiques dans

ce cas.

Les deux points d'intersection sont donc $P(0, 0)$ et $Q\left(2\sqrt{2}, \frac{\pi}{4}\right)$.

ATTENTION

Les coordonnées polaires de la forme (r, θ) où $r = 4 \cos\left(\frac{\pi}{4} + k\pi\right)$ et $\theta = \frac{\pi}{4} + k\pi, k \in \mathbb{Z}$, sont les coordonnées d'un même point.

REMARQUE

Trouver les points d'intersection de deux courbes consiste, en quelque sorte, à résoudre simultanément leurs équations. La résolution simultanée de deux équations polaires ne donnent pas, s'il existe, le point d'intersection au pôle. Il ne faut pas en être surpris, car la coordonnée angulaire θ du pôle est indéterminée. Dans ce cas, s'il existe θ_1 et θ_2 où

$$r = f_1(\theta_1) = 0 \text{ et}$$

$$r = f_2(\theta_2) = 0,$$

alors les deux courbes se rencontrent au pôle.

Initialisons la variable `_EnvAllSolutions` à sa valeur par défaut afin de voir son utilisation dans le prochain exemple.

```
> _EnvAllSolutions:=false;
```

```
    _EnvAllSolutions := false
```

(12.1.5)


Exemple 2

Trouvons les coordonnées des points d'intersection entre la cardioïde $r = 2 + 2 \cos(\theta)$ et le cercle $r = 3$.

```
> C1:=polarplot(2+2*cos(theta),theta=0..2*Pi):
```

```
    C2:=polarplot(3,theta=0..2*Pi,color="Niagara 15");
```

```
> display([C1,C2]);
```



Graphiquement, on observe qu'il y a deux points d'intersection (dans le premier quadrant et dans le quatrième).

```
> Éq_1 := r = 2 + 2*cos(theta);
   Éq_2 := r = 3;
```

$$\dot{Éq}_1 := r = 2 + 2 \cos(\theta)$$

$$\dot{Éq}_2 := r = 3$$

(12.2.1)

```
> solve({Éq_1,Éq_2},{r,theta});
```

$$\left\{ r = 3, \theta = \frac{\pi}{3} \right\}$$

(12.2.2)

Faisons intervenir la variable d'environnement `_EnvAllSolutions` dans la résolution du système précédent.

```
> _EnvAllSolutions := true;
```

`_EnvAllSolutions := true`

(12.2.3)

```
> solve({Éq_1,Éq_2},{r,theta});
```

$$\left\{ r = 3, \theta = -\frac{2}{3} \pi_{_B1} + \frac{1}{3} \pi + 2 \pi_{_Z2} \right\}$$

(12.2.4)

Cette fois, cela s'est avéré utile. La variable système `_B1` est une variable logique (booléenne) prenant les valeurs 0 et 1.

Les deux points d'intersection sont obtenus avec $\theta = \frac{\pi}{3}$ (`_B1 = 0` et `_Z2 = 0`) et avec $\theta = -\frac{\pi}{3}$ (`_B1 = 1` et `_Z2 = 0`).

Dans les deux cas, $r = 3$.

Les points d'intersections sont $P\left(3, -\frac{\pi}{3}\right)$ et $Q\left(3, \frac{\pi}{3}\right)$.


Exemple 3

Trouvons les coordonnées des points d'intersection entre la cardioïde $r = 1 - \sin(\theta)$ et le cercle $r = \sin(\theta)$.

```
> C1 := polarplot(1-sin(theta), theta=0..2*Pi):
```

```
   C2 := polarplot(sin(theta), theta=0..2*Pi, color="Niagara 15");
```

```
> display([C1,C2]);
```



```
> Éq_1:= r=1-sin(theta);
Éq_2:= r=sin(theta);
```

$$\acute{E}q_1 := r = 1 - \sin(\theta)$$

$$\acute{E}q_2 := r = \sin(\theta)$$

(12.3.1)

```
> solve({Éq_1,Éq_2},{r,theta});
```

$$\left\{ r = \sin\left(\frac{2}{3}\pi_{B2} + \frac{1}{6}\pi + 2\pi_{Z3}\right), \theta = \frac{2}{3}\pi_{B2} + \frac{1}{6}\pi + 2\pi_{Z3} \right\}$$

(12.3.2)

Deux points d'intersection sont obtenus avec $\theta = \frac{\pi}{6}$ ($B2 = 0$ et $Z3 = 0$) et avec $\theta = \frac{5\pi}{6}$ ($B2 = 1$ et $Z3 = 0$).

Dans les deux cas, $r = \frac{1}{2}$. En effet

```
> r=sin(Pi/6);
r=sin(5*Pi/6);
```

$$r = \frac{1}{2}$$

$$r = \frac{1}{2}$$

(12.3.3)

Le troisième point d'intersection ne peut être déduit qu'à l'aide du graphique.


Les points d'intersection sont donc $P(0, 0)$, $Q\left(\frac{1}{2}, \frac{\pi}{6}\right)$ et $R\left(\frac{1}{2}, \frac{5\pi}{6}\right)$.

Exemple 4

Trouvons les coordonnées des points d'intersection entre la rosace $r = 4 \cos(2\theta)$ et le cercle $r = 2$.

```
> C1:=polarplot(4*cos(2*theta),theta=0..2*Pi):
C2:=polarplot(2,theta=0..2*Pi,color="Niagara 15"):
```

```
> display([C1,C2]);
```



Trouvons les coordonnées des huit points d'intersection.

```
> Éq_1 := r=4*cos(2*theta);
Éq_2 := r=2;
```

$$\dot{Éq}_1 := r = 4 \cos(2\theta)$$

$$\dot{Éq}_2 := r = 2$$

(12.4.1)

```
> solve({Éq_1,Éq_2},{r,theta});
```

$$\left\{ r = 2, \theta = -\frac{1}{3} \pi_{B3} + \frac{1}{6} \pi + \pi_{Z4} \right\}$$

(12.4.2)

Bien que la variable système `_EnvAllSolutions` soit toujours initialisée à la valeur `true`, l'évaluateur nous révèle seulement quatre points d'intersection, soient avec

$$\theta = \frac{\pi}{6} \quad (_B3 = 0 \text{ et } _Z4 = 0), \quad \theta = -\frac{\pi}{6} \text{ soit } \frac{11\pi}{6} \quad (_B3 = 1 \text{ et } _Z4 = 0),$$

$$\theta = \frac{7\pi}{6} \quad (_B3 = 0 \text{ et } _Z4 = 1), \quad \theta = \frac{5\pi}{6} \quad (_B3 = 1 \text{ et } _Z4 = 1).$$

Pour obtenir les quatres autres points d'intersection, nous devons s'inspirer des tracés des deux graphiques.

En fait, il faut s'inspirer de leurs caractéristiques de symétrie par rapport aux droites $y = \frac{\pi}{4}$ et $y = \frac{3\pi}{4}$.

$$\theta = \frac{\pi}{3}, \quad \theta = \frac{2\pi}{3},$$


$$\theta = \frac{4\pi}{3}, \quad \theta = \frac{5\pi}{3}.$$

Les huit points d'intersection sont: $P\left(2, \frac{\pi}{6}\right), P\left(2, \frac{\pi}{3}\right), P\left(2, \frac{2\pi}{3}\right), P\left(2, \frac{5\pi}{6}\right), P\left(2, \frac{7\pi}{6}\right),$
 $P\left(2, \frac{4\pi}{3}\right), P\left(2, \frac{5\pi}{3}\right)$ et $P\left(2, \frac{11\pi}{6}\right).$

REMARQUE

La difficulté de trouver toutes les solutions découle du fait qu'un même point a une multitude de coordonnées polaires. Il en découle que l'équation polaire d'un lieu peut ne pas être unique. Par exemple, l'équation $r = 4 \cos(2\theta)$ et l'équation $r = -4 \cos(2\theta + \pi)$ représentent le même lieu. En effet,

```
> Lieux_A:=polarplot([2,4*cos(2*theta)],theta=0..2*Pi,  
    color=["Niagara 15","Niagara 4"]):  
Lieux_B:=polarplot([-2,-4*cos(2*theta+Pi)],theta=0..2*Pi,  
    color=["Niagara 15","Niagara 4"]):  
display(Matrix(1,2,[Lieux_A,Lieux_B]));
```



Alors, résolvons simultanément $r = -2$ et $r = -4 \cos(2\theta + \pi)$.


```
> solve({r=-2,r=-4*cos(2*theta+Pi)},{r,theta});  
    {r = -2, theta = -2/3 * pi_B4 + 1/3 * pi + pi_Z5}
```

(12.4.3)

Diverses courbes


Spirale équiangulaire

```
> polarplot(theta/12,theta=0..24*Pi,numpoints=600,axes=None);
```



Spirale logarithmique


```
> polarplot(4*exp(-theta/12),theta=0..18*Pi,numpoints=600,axes=None);
```



Papillon

```
> A:=polarplot(exp(cos(theta))-2*cos(4*theta),theta=0..2*Pi,numpoints=
200,title=typeset("Papillon d'équation ",r=exp(cos(theta))-2*cos(4*
theta))):
B:=polarplot(exp(cos(theta-Pi/2))-2*cos(4*(theta-Pi/2)),theta=-Pi/2.
.3*Pi/2,numpoints=200,title=typeset("Rotation d'un angle de ",Pi/2,"
radians, ce qui donne \n",r=exp(cos(theta-Pi/2))-2*cos(4*(theta-
```


```
Pi/2)))):
display(Matrix(1,2,[A,B]),axes=None,titlefont = ["ROMAN", 15]);
```




Rotation du papillon dans le sens trigonométrique d'un angle de $\frac{\pi}{2}$ radian.

Autres courbes


```
> polarplot(2+7*sin(3*theta),theta=0..2*Pi,numpoints=200,axes=None);
```



```
> polarplot(cos(7*theta/4),theta=0..8*Pi,numpoints=200,axes=None);
```



```
> A:=plot(1+cos(theta/2),theta=0..4*Pi):
B:=polarplot(1+cos(theta/2),theta=0..4*Pi):
display(Matrix(1,2,[A,B]));
```




$r^2 = 4 \cos(\theta)$ et rotation dans le sens trigonométrique d'un angle de $\frac{\pi}{3}$ radian $r^2 = 4 \cos\left(\theta - \frac{\pi}{3}\right)$.

```
> A:=polarplot(sqrt(4*cos(theta)),theta=-Pi/2..Pi/2e):
B:=polarplot(-sqrt(4*cos(theta)),theta=-Pi/2..Pi/2):
```

```

D1:=display(A,B,title=typeset(r^2 = 4*cos(theta)),background=Fond):
A:=polarplot(sqrt(4*cos(theta-Pi/3)),theta=-Pi/6..5*Pi/6):
B:=polarplot(-sqrt(4*cos(theta-Pi/3)),theta=-Pi/6..5*Pi/6):
D2:=display(A,B,title=typeset(r^2 = 4*cos(theta-Pi/3)),background=
Fond):
> display(Matrix(1,2,[D1,D2]));

```



$$r^2 = 4 \sin(2\theta) \text{ et } r^4 = 4 \sin(2\theta)$$

```

> A:=polarplot(sqrt(4*sin(2*theta)),theta=0..Pi/2):
B:=polarplot(-sqrt(4*sin(2*theta)),theta=0..Pi/2):
D1:=display(A,B,title=typeset(r^2 = 4*sin(2*theta)),background=Fond):
A:=polarplot(surd(4*sin(2*(theta)),4),theta=0..Pi/2):
B:=polarplot(-surd(4*sin(2*(theta)),4),theta=0..Pi/2):
D2:=display(A,B,title=typeset(r^4 = 4*sin(2*theta)),background=Fond):
display(Matrix(1,2,[D1,D2]));

```

